

Contents

| | | |
|---|---------------|---|
| 1 | README | 2 |
| 2 | CPU.hdl | 3 |
| 3 | Computer.hdl | 5 |
| 4 | CpuMul.hdl | 6 |
| 5 | ExtendAlu.hdl | 8 |
| 6 | Memory.hdl | 9 |

1 README

```
1  nivkeren,ransha
2  =====
3  Niv Keren, ID 201478351, niv.keren@mail.huji.ac.il
4  Ran Shaham, ID 203781000, ran.shaham1@mail.huji.ac.il
5  =====
6
7          Project 5 - Computer Architecture
8          -----
9
10
11 Submitted Files
12 -----
13 README      - This file.
14 CPU.hdl     - Our implementation of the CPU.
15 ExtendAlu.hdl - ALU with multiplication and shifts.
16 CpuMul.hdl  - CPU that uses the ExtendAlu.
17 Memory.hdl  - Encapsulates 16K RAM, screen memory and keyboard register.
18 Computer.hdl - Combines CPU, Memory and ROM to a working computer!
19
20 Remarks
21 -----
22 * Our implementation followed the instructions given in the lectures, which
23   were simple and bold.
```

2 CPU.hdl

```
1 // This file is part of www.nand2tetris.org
2 // and the book "The Elements of Computing Systems"
3 // by Nisan and Schocken, MIT Press.
4 // File name: projects/05/CPU.hdl
5
6 /**
7  * The Hack CPU (Central Processing unit), consisting of an ALU,
8  * two registers named A and D, and a program counter named PC.
9  * The CPU is designed to fetch and execute instructions written in
10 * the Hack machine language. In particular, functions as follows:
11 * Executes the inputted instruction according to the Hack machine
12 * language specification. The D and A in the language specification
13 * refer to CPU-resident registers, while M refers to the external
14 * memory location addressed by A, i.e. to Memory[A]. The inM input
15 * holds the value of this location. If the current instruction needs
16 * to write a value to M, the value is placed in outM, the address
17 * of the target location is placed in the addressM output, and the
18 * writeM control bit is asserted. (When writeM==0, any value may
19 * appear in outM). The outM and writeM outputs are combinational:
20 * they are affected instantaneously by the execution of the current
21 * instruction. The addressM and pc outputs are clocked: although they
22 * are affected by the execution of the current instruction, they commit
23 * to their new values only in the next time step. If reset==1 then the
24 * CPU jumps to address 0 (i.e. pc is set to 0 in next time step) rather
25 * than to the address resulting from executing the current instruction.
26 */
27
28 CHIP CPU {
29
30     IN  inM[16],          // M value input  (M = contents of RAM[A])
31         instruction[16], // Instruction for execution
32         reset;           // Signals whether to re-start the current
33                          // program (reset==1) or continue executing
34                          // the current program (reset==0).
35
36     OUT outM[16],         // M value output
37         writeM,           // Write to M?
38         addressM[15],     // Address in data memory (of M)
39         pc[15];           // address of next instruction
40
41     PARTS:
42         // C-Instruction decoding
43         And16(a=instruction, b=true, out[15]=opcode, out[12]=ca,
44             out[11]=c1, out[10]=c2, out[9]=c3, out[8]=c4, out[7]=c5,
45             out[6]=c6, out[5]=writeA, out[4]=writeD, out[3]=writeM1,
46             out[2]=j1, out[1]=j2, out[0]=j3);
47
48         // writeM is true iff it is a C-instruction and d3==1
49         And(a=writeM1, b=opcode, out=writeM);
50
51         // A Register logic
52         Mux16(a=instruction, b=aluout, sel=opcode, out=aregin);
53         Not(in=opcode, out=negopcode);
54         // Load address to ARegister if it is an A-Instruction or d1==1
55         Or(a=negopcode, b=writeA, out=aload);
56         ARegister(in=aregin, load=aload, out=aregout, out[0..14]=addressM);
57
58         // D Register logic - similar to A Register
59         And(a=writeD, b=opcode, out=dload);
```

```

60         DRegister(in=aluout, load=dload, out=dregout);
61
62     // ALU logic
63     // The a-bit in the C-Instruction determines whether to handle
64     // A value or M value. This bit is extracted in the C-Instruction
65     // decoding at the beginning of the chip
66     Mux16(a=aregout, b=inM, sel=ca, out=aorm);
67     ALU(x=dregout, y=aorm, zx=c1, nx=c2, zy=c3, ny=c4, f=c5, no=c6,
68         zr=zr, ng=ng, out=aluout, out=outM);
69
70     // PC logic
71     Not(in=ng, out=nng);
72     Not(in=zr, out=nzr);
73     And(a=nng, b=nzr, out=pt);
74
75     And(a=j3, b=pt, out=jgt);
76     And(a=j1, b=ng, out=jlt);
77     And(a=j2, b=zr, out=jeq);
78     And(a=j1, b=j2, out=tmp);
79     And(a=tmp, b=j3, out=jmp);
80     // This covers every jump condition (e.g jge is satisfied by jgt or
81     // jeq)
82     Or8Way(in[0]=jmp, in[1]=jgt, in[2]=jlt, in[3]=jeq, in[4..7]=false,
83         out=pcload1);
84     And(a=pcload1, b=opcode, out=pcload);
85
86     PC(in=aregout, load=pcload, inc=true,
87         reset=reset, out[0..14]=pc);
88 }

```

3 Computer.hdl

```
1 // This file is part of www.nand2tetris.org
2 // and the book "The Elements of Computing Systems"
3 // by Nisan and Schocken, MIT Press.
4 // File name: projects/05/Computer.hdl
5
6 /**
7  * The HACK computer, including CPU, ROM and RAM.
8  * When reset is 0, the program stored in the computer's ROM executes.
9  * When reset is 1, the execution of the program restarts.
10 * Thus, to start a program's execution, reset must be pushed "up" (1)
11 * and "down" (0). From this point onward the user is at the mercy of
12 * the software. In particular, depending on the program's code, the
13 * screen may show some output and the user may be able to interact
14 * with the computer via the keyboard.
15 */
16
17 CHIP Computer {
18
19     IN reset;
20
21     PARTS:
22         // Exactly as shown in class: Unit 5.4, 25'40''
23         CPU(inM=memout, instruction=romout, reset=reset, writeM=writeM,
24            outM=outM, addressM=addressM, pc=pc);
25         ROM32K(address=pc, out=romout);
26         Memory(in=outM, address=addressM, load=writeM, out=memout);
27 }
```

4 CpuMul.hdl

```
1  /**
2  * This chip is an extension of the book CPU by using the extended ALU.
3  * More specifically if instruction[15]==0 or (instruction[14] and instruction[13] equals 1)
4  * the CpuMul behave exactly as the book CPU.
5  * While if it is C instruction and instruction[13] == 0 the output will be D*A/M
6  * (according to instruction[12]).
7  * Moreover, if it is c instruction and instruction[14] == 0 it will behave as follows:
8  *
9  * instruction: | 12 | 11 | 10 |
10 * -----
11 * shift left D | 0 | 1 | 1 |
12 * shift left A | 0 | 1 | 0 |
13 * shift left M | 1 | 1 | 0 |
14 * shift right D | 0 | 0 | 1 |
15 * shift right A | 0 | 0 | 0 |
16 * shift right M | 1 | 0 | 0 |
17 */
18
19 CHIP CpuMul{
20
21     IN  inM[16],          // M value input  (M = contents of RAM[A])
22         instruction[16], // Instruction for execution
23         reset;           // Signals whether to re-start the current
24                           // program (reset=1) or continue executing
25                           // the current program (reset=0).
26
27     OUT outM[16],         // M value output
28         writeM,           // Write into M?
29         addressM[15],     // Address in data memory (of M)
30         pc[15];          // address of next instruction
31
32     PARTS:
33     // The same as CPU except for the ALU part, which is more intuitive.
34
35     // C-Instruction decoding
36     And16(a=instruction, b=true, out[15]=opcode, out[12]=ca,
37           out[5]=writeA, out[4]=writeD, out[3]=writeM1,
38           out[2]=j1, out[1]=j2, out[0]=j3);
39
40     // writeM is true iff it is a C-instruction and d3==1
41     And(a=writeM1, b=opcode, out=writeM);
42
43     // A Register logic
44     Mux16(a=instruction, b=aluout, sel=opcode, out=aregin);
45     Not(in=opcode, out=negopcode);
46     // Load address to ARegister if it is an A-Instruction or d1==1
47     Or(a=negopcode, b=writeA, out=aload);
48     ARegister(in=aregin, load=aload, out=aregout, out[0..14]=addressM);
49
50     // D Register logic - similar to A Register
51     And(a=writeD, b=opcode, out=dload);
52     DRegister(in=aluout, load=dload, out=dregout);
53
54     // ALU logic
55     // The a-bit in the C-Instruction determines whether to handle
56     // A value or M value. This bit is extracted in the C-Instruction
57     // decoding at the beginning of the chip
58     Mux16(a=aregout, b=inM, sel=ca, out=aorm);
59     ExtendAlu(x=dregout, y=aorm, instruction=instruction[6..14],
```

```

60         zr=zr, ng=ng, out=aluout, out=outM);
61
62     // PC logic
63     Not(in=ng, out=nng);
64     Not(in=zr, out=nzr);
65     And(a=nng, b=nzr, out=pt);
66
67     And(a=j3, b=pt, out=jgt);
68     And(a=j1, b=ng, out=jlt);
69     And(a=j2, b=zr, out=jeq);
70     And(a=j1, b=j2, out=tmp);
71     And(a=tmp, b=j3, out=jmp);
72     // This covers every jump condition (e.g jge is satisfied by jgt or
73     // jeq)
74     Or8Way(in[0]=jmp, in[1]=jgt, in[2]=jlt, in[3]=jeq, in[4..7]=false,
75            out=pcload1);
76     And(a=pcload1, b=opcode, out=pcload);
77
78     PC(in=aregout, load=pcload, inc=true,
79        reset=reset, out[0..14]=pc);
80 }

```

5 ExtendAlu.hdl

```
1  /**
2  * The input of the extends ALU is instruction[9] and x[16],y[16].
3  * the output is define as follows:
4  * If instruction[7..8] equals 1 the the output is exactly as the ALU.
5  * Where instruction[5]=zx,instruction[4]=nx,...,instruction[0]=no.
6  * If instruction[7] equals 0 the output will be x*y and disregard the rest
7  * of the instruction.
8  *
9  * If instruction[8] equals 0 the output will be shift.
10 * Then, if instruction[4] equals 0 it will return shift of y otherwise shift
11 * of x, moreover if instruction[5] equals 0 it will return shift right
12 * otherwise shift left.
13 */
14 CHIP ExtendAlu{
15     IN x[16],y[16],instruction[9];
16     OUT out[16],zr,ng;
17
18     PARTS:
19         ALU(x=x, y=y, zx=instruction[5], nx=instruction[4],
20             zy=instruction[3], ny=instruction[2],
21             f =instruction[1], no=instruction[0],
22             out=aluout);
23
24     // Compute all shifts of x,y
25     ShiftLeft(in=x, out=shiftxl);
26     ShiftLeft(in=y, out=shiftyl);
27     ShiftRight(in=x, out=shiftxr);
28     ShiftRight(in=y, out=shiftyr);
29     // Determine which shift to choose according to instruction[4..5]
30     Mux4Way16(a=shiftyr, b=shiftxr, c=shiftyl, d=shiftxl,
31              sel=instruction[4..5], out=shiftout);
32
33     // Compute x*y
34     Mul(a=x, b=y, out=mulout);
35
36     // Determine the output according to instruction[7..8]
37     Mux4Way16(a=mulout, b=shiftout, c=mulout, d=aluout,
38              sel=instruction[7..8], out[15]=ng, out[0..7]=out07,
39              out[8..15]=out815, out=out);
40
41     // Same as ng check in the ALU
42     Or8Way(in=out07, out=zr1);
43     Or8Way(in=out815, out=zr2);
44     Or(a=zr1, b=zr2, out=nzr);
45     Not(in=nzr, out=zr);
46 }
```


6 Memory.hdl

```
1 // This file is part of www.nand2tetris.org
2 // and the book "The Elements of Computing Systems"
3 // by Nisan and Schocken, MIT Press.
4 // File name: projects/05/Memory.hdl
5
6 /**
7  * The complete address space of the Hack computer's memory,
8  * including RAM and memory-mapped I/O.
9  * The chip facilitates read and write operations, as follows:
10  *   Read: out(t) = Memory[address(t)](t)
11  *   Write: if load(t-1) then Memory[address(t-1)](t) = in(t-1)
12  * In words: the chip always outputs the value stored at the memory
13  * location specified by address. If load==1, the in value is loaded
14  * into the memory location specified by address. This value becomes
15  * available through the out output from the next time step onward.
16  * Address space rules:
17  * Only the upper 16K+8K+1 words of the Memory chip are used.
18  * Access to address>0x6000 is invalid. Access to any address in
19  * the range 0x4000-0x5FFF results in accessing the screen memory
20  * map. Access to address 0x6000 results in accessing the keyboard
21  * memory map. The behavior in these addresses is described in the
22  * Screen and Keyboard chip specifications given in the book.
23  */
24
25 CHIP Memory {
26     IN in[16], load, address[15];
27     OUT out[16];
28
29     PARTS:
30         // The two MSBs determine which memory block to load to
31         DMux4Way(in=true, sel=address[13..14], a=ramload1, b=ramload2
32             , c=screenload1, d=ignore);
33         Or(a=ramload1, b=ramload2, out=ramload3);
34
35         // We only want to load a memory block if it is the chosen one by
36         // the MSBs and load input == 1
37         And(a=load, b=ramload3, out=ramload);
38         RAM16K(in=in, address=address[0..13], load=ramload, out=ramout);
39
40         And(a=load, b=screenload1, out=screenload);
41         Screen(in=in, address=address[0..12], load=screenload, out=screenout);
42
43         Keyboard(out=kbdout);
44
45         // The MSBs also determine from which memory block to output
46         Mux4Way16(a=ramout, b=ramout, c=screenout, d=kbdout,
47             sel=address[13..14], out=out);
48 }
```